

_____という接続 ($\backslash a \rightarrow k a c$) を m に渡す。 m は最後にこの接続を呼び出すのが普通だが、無視したり、他の接続を呼び出ししたりすることも可能である。これが、ジャンプなどの命令に対応する。

6.2 UtilCont – 接続の導入

Util に `break`, `continue` などを導入するために、`Expr` の定義に次のように構子を追加する。また、`goto` 文を導入するため、ラベルも導入する。

ファイル `ContType.hs`

```

1  data Expr = Const Target | Var String
2          | If Expr Expr Expr | While Expr Expr
3          | Let [Decl] Expr | Val Decl Expr
4          | Lambda String Expr | Delay Expr | App Expr Expr
5          -- ここまでは、Utl1と同じ
6          | Begin [LabeledExpr]      -- ブロック
7          | Break                    -- break 文
8          | Continue                 -- continue 文
9          | Goto String              -- goto 文
10         deriving Show
11  type LabeledExpr = (Maybe String, Expr) -- ラベル付きの式

```

これに対する具象構文としては、

```

Expr          → ... | begin LabeledExprSeq
              | break | continue | goto Var
LabeledExprSeq → LabeledExpr end | LabeledExpr ; LabeledExprSeq
LabeledExpr   → Expr | Var : Expr

```

を想定する。

実際の `UtilCont` では接続とともに状態や入出力も扱いたいので、計算のモナドは、 K そのものではなく、“状態の変化”を“結果”として持つ K (`ST (WithIO s) v`) a (と同型の型) とする。ここで、 s は状態の型である。

ファイル `Cont.hs`

```

1  newtype KST v s a = KST ((a -> WithIO s -> (v,WithIO s))
2                          -> WithIO s -> (v,WithIO s))
3
4  unKST :: KST v s a -> (a -> WithIO s -> (v,WithIO s))
5          -> WithIO s -> (v,WithIO s)
6  unKST (KST m) = m

```

この `KST` の定義にあわせて、`set` など状態に関する関数も書き直しておく。

ファイル `Cont.hs`

```

1  instance MyState (KST v) where
2    get p = KST (\ c (s,i,o) -> c (fst (p s)) (s,i,o))
3    set p v = KST (\ c (s,i,o) -> c () (snd (p s) v,i,o))
4

```

```

5 instance MyStream (KST v s) where
6   readChar = KST (\ c (s,ch:i,o) -> c ch (s,i,o))
7   eof      = KST (\ c (s,i,o) -> c (null i) (s,i,o))
8   writeStr v = KST (\ c (s,i,o) -> c () (s,i,o ++ v))
9
10 abort :: (WithIO s -> (v,WithIO s)) -> KST v s a
11 abort r = KST (_____)
```

すると、set p v は状態の p で表される位置に v をセットし、() と新しい状態を接続に渡す。また、abort r は現在の接続を無視して r という値を全体の計算の結果としている。これは計算を途中で中止することに相当する。

Const, Var, Let などに対しては comp は変更する必要はない。変更された部分のうち、Goto, Break, Continue に対する comp の定義は以下ようになる。

ファイル ContCompiler.hs

```

1 comp (Goto l)          = mkGoto l "()"
2 comp Break            = mkGoto "_break" "()"
3 comp Continue         = mkGoto "_while" "_break"
4
5 mkGoto l v = TApp1 (TVar "abort") (TApp1 (TVar l) (TVar v))
```

ここで goto, break, continue について、変換前と変換後をそれぞれ Util と Haskell の文法で記述すると次の表になる。

ソース (Util)	ターゲット (Haskell)
goto label	abort (label ())
break	abort (_break ())
continue	abort (_while _break)

一方 goto label, break はそれぞれ、現在の接続は無視して、label, _break という識別子に束縛されている接続を起動する式に翻訳される。これが“ジャンプ”に相当する。また continue も、現在の接続は無視して、_while という識別子に束縛されている計算に _break という接続を渡す。

ところで while ~ do ~ に対しては、break に対応する接続を変数に格納する必要があるため、定義がやや複雑になる。

ファイル ContCompiler.hs

```

1 comp (While e1 e2)    = compWhile e1 e2
2
3 compWhile e1 e2 = TApp1 (TVar "KST")
4   (TLambda1 "_break"
5     (TLet [(PVar "_while", TApp1 (TVar "unKST") body)]
6           (TApp1 (TVar "_while") (TVar "_break")))))
7   where body = comp e1 'TBind' TLambda1 "_b"
8             (TIf (TVar "_b") (comp e2 'TBind' TLambda1 "_"
9                               (TApp1 (TVar "KST") (TVar "_while")))
10              (TReturn (TVar "()")))
```

ソース (Util)	ターゲット (Haskell)
<code>while c do t</code>	<pre>KST (\ _break -> let KST _while = c >>= \ _b -> if _b then i >>= \ _ -> KST _while else return () in _while _break)</pre>

ここで、`_break` は _____ を表す接続で、`_while _break` は _____ を表す接続である。これらの接続が、それぞれ `break`, `continue` に対応する。

例えば、UtilCont プログラム (右は対応する C プログラム) :

<pre>1 foo = \ y -> begin 2 set xP 1; set yP y; 3 while get yP > 0 do begin 4 val x = get xP in 5 val y = get yP in 6 if y == 10 then break 7 else if y == 3 then begin 8 set yP (y - 1); continue 9 end else (); 10 set xP (x * y); 11 set yP (y - 1) 12 end; 13 get xP 14 end</pre>	<pre>1 int foo(int y) { 2 int x = 1; 3 while (y > 0) { 4 5 if (y == 10) break; 6 else if (y == 3) { 7 y--; continue; 8 } 9 x = x * y; 10 y--; 11 } 12 return x; 13 }</pre>
----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

をコンパイルすると、次の Haskell プログラムが得られる。

```
1  foo = \ y ->
2    set xP 1 >>= \ _ ->
3    set yP y >>= \ _ ->
4    KST (\ _break ->
5      let KST _while
6        = get yP >>= \ y ->
7          if y > 0 then
8            get xP >>= \ x ->
9            get yP >>= \ y ->
10           (if y == 10 then abort (_break ()) else
11            if y == 3 then
12              set yP (y - 1) >>= \ _ ->
13              abort (_while _break)
14            else return ()) >>= \ _ ->
15           set xP (x * y) >>= \ _ ->
16           set yP (y - 1) >>= \ _ ->
17           KST _while
18            else return ()
19      in _while _break) >>= \ _ ->
20    get xP
```

この foo の型は Integer -> KST a (Integer,Integer) a Integer であるから、値を取り出すためには、整数と初期接続 (通常、\ a s -> (a,s))、初期状態 ((0,0), "", "") を渡す必要がある。ここで

```
1 evalKST st s = fst (unKST st (\ a s -> (a,s)) (s, "", ""))
```

と定義すると、evalKST (foo 9) (0,0) の結果は、_____ に、evalKST (foo 11) (0,0) は __ になる。(参考: 9! = 362880)

さらに goto に対する意味を与えるためには、ブロック (begin~end) のなかで、“ラベル”に適切な接続を与える必要があるが、Begin に対する comp の定義は長くなってしまっているので、ここに示さず、変換前と変換後の形の例のみを示す。

ソース (Util)	ターゲット (Haskell)
begin	KST (\ _end -> let
l bl 1: s ₁	l bl 1 = \ _ -> unKST s ₁ l bl 2
l bl 2: s ₂	l bl 2 = \ _ -> unKST s ₂ l bl 3
l bl 3: s ₃	l bl 3 = \ _ -> unKST s ₃ _end
end	in l bl 1 (())

この s₁, s₂, s₃ の中には、goto l**bl**1, goto l**bl**2, goto l**bl**3 が含まれているかもしれない。ターゲット (Haskell) プログラム中の識別子 l**bl**1, l**bl**2, l**bl**3 に束縛されているのはそれぞれ、同名のラベル l**bl**1, l**bl**2, l**bl**3 に対応する接続である。

例えば、次の UtilCont プログラム (右は対応する C プログラム) :

<pre>1 bar = \ _ -> begin 2 set xP 1; 3 label1: 4 if get xP > 100 then goto label2 5 else (); 6 set xP (get xP * 2); 7 goto label1; 8 label2: 9 get xP 10 end</pre>	<pre>1 int bar(void) { 2 int x = 1; 3 label1: 4 if (x > 100) 5 goto label2; 6 x = x * 2; 7 goto label1; 8 label2: 9 return x; 10 }</pre>
----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------	-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------

は次のような Haskell プログラムにコンパイルされる。

```
1 bar = \ _ ->
2   KST (\ _end ->
3     let label1 = \ _ -> unKST
4       (get xP >>= \ x ->
5         (if x > 100 then abort (label2 ())
6           else return ())) >>= \ _ ->
7         get xP >>= \ x ->
8         set xP (x * 2) >>= \ _ ->
9         abort (label1 ())) label2
10    label2 = \ _ -> unKST (get xP) _end
11    in unKST (set xP 1) label1)
```

そして、evalKST (bar ()) (0,0) を評価すると、結果は _____ になる。

問 6.2.1 次の C の関数とほぼ同等な Haskell の関数をモナドを用いて作成せよ。

```
1  int hoge(int n) {
2      int i = 1, sum = 0;
3      while (i <= n) {
4          sum = sum + i;
5          if (sum > 21) {
6              sum = 0;
7              break;
8          }
9          i = i + 1;
10     }
11     return sum;
12 }
```

ごちゃごちゃと goto 文を多用したスパゲッティコードは、結局、相互再帰を多用することと同等であり、プログラムの意味がわかりにくくなる。

6.3 callcc とは

ここで紹介する callcc は Scheme や Ruby などが採用している、プログラマが接続を直接操作することができるプリミティブである。Scheme では call-with-current-continuation または、省略して call/cc と書く。(Scheme では “-” も “/” も関数名の一部として使えることに注意する。)

1 引数の関数 *thunk* に対して、callcc *thunk* のように使用すると _____ を引数として、*thunk* を呼び出す。*thunk* のなかで、この接続を呼び出せば、呼出しの接続は無視されて (= ジャンプして) callcc が呼ばれたときの接続にその値が返される。一方 *thunk* が接続を呼び出さなければ、*thunk* 自身の戻り値が callcc 式全体の戻り値になる。まず、トリビアルな例として、

```
1  baz = \ x ->
2      callcc (\ k ->
3          100 + (if x == 0 then 1 else k x))
```

という関数を考える。ここで baz 0 を評価すると普通に足し算が計算され、値は ___ になる。一方、baz 1 の場合は、接続 k が呼び出されるので 100 を足す部分はスキップされて、戻り値は _ となる。

また callcc のよくある使い方は、try ~ catch と同じような大域脱出である。

```
1  multlist = \ xs ->
2      let aux = \ xs -> \ k -> begin
3          set xP 1; set yP xs;
4          while not (null (get yP)) do
5              val y = get yP in val n = head y in
6              if n == 0 then k 0 else
7                  begin set xP (get xP * n); set yP (tail y);
8                      writeStr "_";
```

```

9           write n
10          end;
11          get xP
12        end in
13    val result = callcc (\ k -> aux xs k) in begin
14        writeStr "_result_=";
15        write result
16    end

```

この関数はリストの要素の掛け算を求める。要素の中に0が見つかったら、大域脱出して multlist 全体の値は _ になる。

このような大域脱出だけならば、言語の仕様に callcc のような大がかりな仕掛けをいれておく必要はない。callcc の本当の価値はコルーチンなどの普通でない制御構造を実現できるところにある。

6.4 コルーチン (coroutine)

コルーチンとは、2 つ以上のプログラムの実行単位が、同時にながら実行されていく方式のことである。サブルーチン (subroutine) のように、実行単位の間には主と副といった従属関係はなく、コルーチンを構成する個々のルーチンは互いに対等な関係である。

例えば、

```

1    increase n k =
2        if n > 10 then ()
3        else begin writeStr "_i:"; write n;
4            increase (n + 1) (callcc k) end;
5    decrease n k =
6        if n < 0 then ()
7        else begin writeStr "_d:"; write n;
8            decrease (n - 1) (callcc k) end

```

という2つの関数を定義して

```
increase 0 (decrease 10)
```

という式を実行すると、

と出力される。

注意: 実は、この Util プログラムを Haskell に変換すると型エラーになり、そのままではコンパイル・実行できない。これは、callcc の引数の型と戻り値の型が同一になる必要があるからである。いくつかトリッキーな変換をすると、意味を変えずに型付け可能な定義に書き換えが可能で、上記のような実行結果になる。しかし、ここはコルーチ

ンのアイデアを説明するのが本旨なので、型付けをするためのトリックの詳細には立ち入らないことにする。

なお、Scheme では、

```
1 (define (increase n k)
2   (if (> n 10) '()
3       (begin (display "_i:") (display n)
4               (increase (+ n 1) (call/cc k))))))
5 (define (decrease n k)
6   (if (< n 0) '()
7       (begin (display "_d:") (display n)
8               (decrease (- n 1) (call/cc k))))))
```

のように対応する関数を定義して

```
(increase 0 (lambda (k) (decrease 10 k)))
```

という式を実行すると上記の出力が得られる。

ここでは increase と decrease という 2 つの関数が交互に実行されていることがわかる。スレッドと似ているが、2 つのルーチンが同時に実行される訳ではない。

また callcc は、コルーチンの他にこれまでに紹介したエラー処理 (try ~ catch) や非決定性などのプリミティブも、callcc を用いて定義できることがわかっている。ある意味でオールマイティなプリミティブである。しかし、その詳細の解説については、ここでは割愛する。

6.5 callcc の表現

我々の言語 UtilCont に callcc を導入するには、接続を関数として渡すためのコードを用意すれば良い。callcc に対応する関数の定義は次のようになる。ファイル Cont.hs

```
1 callcc :: ((a -> KST v s b) -> KST v s a) -> KST v s a
2 callcc h = KST (\ c -> let k a = KST (\ d -> c a)
3                       in unKST (h k) c)
4 -- KST, unKST がなければ
5 -- callcc h = \ c -> let k a = \ d -> c a
6 --                       in h k c
```

この callcc の定義中で用いられている k は現在の接続 (d) を捨て、キャプチャされた接続 (c) を呼び出すという関数である。

コンパイラーは単に callcc という名前の UtilCont の関数を Haskell の callcc にコンパイルすれば良い。

ソース (Util)	ターゲット (Haskell)
callcc m	m >>= \ _x -> callcc _x

また、head, tail, null, not, show などの 1 引数で副作用を持たない関数は、次のようにコンパイルされる。

ソース (Util)	ターゲット (Haskell)
<code>funWithOneArg m</code>	<code>m >>= \ _x -> return (funWithOneArg _x)</code>

例えば、先に紹介した UtilCont プログラム `multlist` をコンパイルすると、次の Haskell プログラムが得られる。

```

1 multlist = \ xs ->
2   let aux = \ xs ->
3     return (\ k ->
4       set xP 1          >>= \ _ ->
5       set yP xs         >>= \ _ ->
6       KST (\ _break ->
7         let KST _while
8           = get yP          >>= \ y ->
9             if not (null y) then
10              get yP          >>= \ y ->
11              (if head y == 0 then k 0 else
12               get xP          >>= \ x ->
13                set xP (x * head y)
14                  >>= \ _ ->
15                  set yP (tail y) >>= \ _ ->
16                  writeStr "␣" >>= \ _ ->
17                  write (head y)) >>= \ _ ->
18                KST _while
19              else return ()
20              in _while _break) >>= \ _ ->
21         get xP)
22   in callcc (\ k -> aux xs >>= \ _f ->
23     _f k) >>= \ result ->
24   writeStr ";␣result␣=" >>= \ _ ->
25   write result

```

このとき

```

evalKSTIO st s =
  let (_,(_,_,o)) = unKST st (\ a s -> (a,s)) (s,"","")
  in o

```

と定義すると、`evalKSTIO (multlist [1,2,3,4,5]) (0, [])` は、“ 1 2 3 4 5; result = 120” を表示し、

一方、`evalKSTIO (multlist [1,2,3,0,4,5]) (0, [])` は“ 1 2 3; result = 0” となる。つまり、0 が現れた時点で乗算を打ち切っていることがわかる。

6.6 さらに詳しく知りたい人のために...

接続に関する文献は数多くあるが、[1] は接続の「発見」について、振り返っている珍しいものである。文献 [2] は、`call/cc` がある意味で「オールマイティ」であることについての説明を与えている。文献 [3] は、Java などの命令型言語に、

call/cc のような接続を扱うオペレータを導入する方法を述べている。文献 [4] は mfixU などの不動点演算子について解説している。

この章の参考文献

- [1] John C. Reynolds, “The Discoveries of Continuations” *Lisp and Symbolic Computation*, 6, (233–247). 1993 年
- [2] Andrzej Filinski, “Representing Monads” *21st ACM Symposium on Principles of Programming Languages*. 1994 年
- [3] T. Sekiguchi, T. Sakamoto, and A. Yonezawa, “Portable Implementation of Continuation Operators in Imperative Languages by Exception Handling” *Advances in Exception Handling Techniques*. Springer-Verlag, LNCS 2022. 2001 年 <http://www.yl.is.s.u-tokyo.ac.jp/amo/>
- [4] Levent Erkök, and John Launchbury, “Recursive Monadic Bindings” *Proc. of the International Conference on Functional Programming*. 2000 年